Laboratory 2: Variational Calculus

Integrals

The function **int** computes an indefinite or definite integral of the expression **expr** with respect to the variable **x**. The name **integrate** is a synonym for **int**. Indefinite integration is performed if the second argument **x** is a name. Note that no constant of integration appears in the result.

int(**expr**, **x**)

Definite integration is performed if the second argument is of the form **x=a..b** where **a** and **b** are the endpoints of the interval of integration.

int(**expr**, **x=a..b, ...**)

```
> int( sin(x), x );
```

$$-\cos(x)$$

```
> int( cos(x), x=0..Pi );
```

$$0$$

If Maple cannot find a closed form expression for the integral, the function call itself is returned.

```
> int( exp(-x^3)/(x^2+1), x = 0..1 );
```

$$\int_0^1 \frac{e^{(-x^3)}}{x^2+1}\,dx$$

The most common command for numerical integration is **evalf(Int(f, x=a..b))** where the integration command is expressed in inert form to avoid first invoking the symbolic integration routines. It is also possible to invoke **evalf** on an unevaluated integral returned by the symbolic **int** command, as in **evalf(int(f, x=a..b))**, if it happens that symbolic **int** fails (returns an unevaluated integral).

```
> evalf(int( exp(-x^3)/(x^2+1), x = 0..1 ));
```

$$0.6649369431$$

```
> evalf(Int( exp(-x^3)/(x^2+1), x = 0..1 ));
```

$$0.6649369431$$

```
>
```

The definition of the integral functional

The variational calculus studies the extremal (minimal/maximal) points for the integral functional

$$J(y(x)) = \int_a^b L\left(x, y(x), \frac{d}{dx}y(x)\right) dx$$

where the function L(x,y,y') is called the Lagrangian of the functional.

```
> restart:
```

Let consider the integral functional $J(y(x)) = \int_0^{\pi} y(x)^2 \left(1 - \left(\frac{d}{dx} y(x)\right)^2\right) dx$ and evaluate $J(y(x))$ for

$y(x) = x^2$

First we define the lagrangian as a function

```
> L:=(x,u,v)->u^2*(1-v^2);
```

$$L := (x, u, v) \to u^2 (1 - v^2)$$

We define the functional in which y is an expresion depending on x

```
> J:=y->int(L(x,y,diff(y,x)),x=0..Pi);
```

$$J := y \to \int_0^{\pi} L\left(x, y, \frac{d}{dx} y\right) dx$$

Now, we define the function $y(x) = x^2$

```
> y1:=x->x^2;
```

$$y1 := x \to x^2$$

To evaluate $J(y(x))$ for $y(x) = x^2$ just call $J(y1(x))$

```
> J(y1(x));
```

$$-\frac{4}{7}\pi^7 + \frac{1}{5}\pi^5$$

```
> evalf(%);
```

$$-1664.677909$$

The first order variation of the integral functional

The first order variation of the integral functional J in $y(x)$ that corresponds to the direction $u(x)$ is the application defined by

$$\delta J(y(x), u(x)) = \lim_{\lambda \to 0} \frac{J(y(x) + \lambda\, u(x)) - J(y(x))}{\lambda}$$

or

$\delta J(y(x), u(x)) = \frac{\partial}{\partial \lambda} J(y(x) + \lambda\, u(x))$ evaluated for $\lambda = 0$.

Lets calculate the first variation of the functional $J(y(x)) = \int_{-1}^{1} \frac{d}{dx} y(x)\, dx$ in some arbitrary function $y(x)$.

First we define the lagrangian and after that we evaluate the limit.

```
> L:=(x,u,v)->v;
```

$$L := (x, u, v) \to v$$

Now we define the functional

```
> J:=y->int(L(x,y,diff(y,x)),x=-1..1);
```

$$J := y \rightarrow \int_{-1}^{1} \mathrm{L}\left(x, y, \frac{d}{dx} y\right) dx$$

Lets calculate the first order variation using its first definition

```
> delta:=limit((J(y(x)+lambda*u(x))-J(y(x)))/lambda,lambda=0);
```

$$\delta := \lim_{\lambda \to 0} \frac{\displaystyle\int_{-1}^{1} \left(\frac{d}{dx} \mathrm{y}(x)\right) + \lambda \left(\frac{d}{dx} \mathrm{u}(x)\right) dx - \int_{-1}^{1} \frac{d}{dx} \mathrm{y}(x)\, dx}{\lambda}$$

To write these two integrals in one we have to use **combine** command

```
> delta:=combine(delta);
```

$$\delta := \lim_{\lambda \to 0} \int_{-1}^{1} \frac{d}{dx} \mathrm{u}(x)\, dx$$

Notice that the limit expresion does not depend on λ , so we use **simplify** comand in order to drop λ

```
> delta:=simplify(delta);
```

$$\delta := \int_{-1}^{1} \frac{d}{dx} \mathrm{u}(x)\, dx$$

We know that the answer should be $\mathrm{u}(1) - \mathrm{u}(-1)$, to force MAPLE to give this answer we have to use *integration by parts command* included in the *student package*. (See help for details)

```
> with(student):
> intparts(delta, 1);
```

$$\mathrm{u}(1) - \mathrm{u}(-1) - \int_{-1}^{1} 0\, dx$$

```
> simplify(%);
```

$$\mathrm{u}(1) - \mathrm{u}(-1)$$

Since $\mathrm{u}(1) = \mathrm{u}(-1)$ $= 0$ then the first variation of this functional is in fact 0.
Now, lets compute the first variation using the seccond definition

```
> e1:=J(y(x)+lambda*u(x));
```

$$e1 := \int_{-1}^{1} \left(\frac{d}{dx} \mathrm{y}(x)\right) + \lambda \left(\frac{d}{dx} \mathrm{u}(x)\right) dx$$

```
> e2:=diff(e1,lambda);
```

$$e2 := \int_{-1}^{1} \frac{d}{dx} \mathrm{u}(x)\, dx$$

Notice that $e2$ does not depend on λ, so the next command is not necessary in this case, but usualy the expression depends on lambda. In these cases we have to use next line in order to evaluate $e2$ for $\lambda = 0$.

```
> delta:=subs(lambda=0,e2);
```

$$\delta := \int_{-1}^{1} \frac{d}{dx} \mathrm{u}(x)\, dx$$

```
> delta:=intparts(delta,1);
```

$$\delta := \mathrm{u}(1) - \mathrm{u}(-1) - \int_{-1}^{1} 0\, dx$$

```
> delta:=simplify(delta);
```

$$\delta := \mathrm{u}(1) - \mathrm{u}(-1)$$

The second order variation of the functional

The second order variation of the integral functional J in $\mathrm{y}(x)$ that corresponds to the direction $\mathrm{u}(x)$ is the application defined by

$\delta^2\, \mathrm{J}(\mathrm{y}(x), \mathrm{u}(x)) = \frac{\partial^2}{\partial \lambda^2} \mathrm{J}(\mathrm{y}(x) + \lambda\, \mathrm{u}(x))$ evaluated for $\lambda = 0$.

Lets calculate the second variation of the functional $\mathrm{J}(\mathrm{y}(x)) = \int_{-1}^{1} \mathrm{y}(x)^2 \left(\frac{d}{dx} \mathrm{y}(x)\right) dx$ in the function

$\mathrm{y}(x) = x^2$. First we define the lagrangian:

```
> L:=(x,u,v)->u^2*v;
```

$$L := (x, u, v) \to u^2\, v$$

Now we define the functional:

```
> J:=y->int(L(x,y,diff(y,x)),x=-1..1);
```

$$J := y \to \int_{-1}^{1} \mathrm{L}\left(x, y, \frac{d}{dx} y\right) dx$$

```
> y1:=x->x^2;
```

$$y1 := x \to x^2$$

```
> e1:=J(y1(x)+lambda*u(x));
```

$$e1 := \int_{-1}^{1} (x^2 + \lambda\, \mathrm{u}(x))^2 \left(2\, x + \lambda \left(\frac{d}{dx} \mathrm{u}(x)\right)\right) dx$$

```
> e2:=diff(e1,lambda$2);
```

$$e2 := \int_{-1}^{1} 2\,u(x)^2 \left(2\,x + \lambda \left(\frac{d}{dx} u(x)\right)\right) + 4\,(x^2 + \lambda\,u(x)) \left(\frac{d}{dx} u(x)\right) u(x)\,dx$$

```
> e3:=subs(lambda=0,e2);
```

$$e3 := \int_{-1}^{1} 4\,u(x)^2\,x + 4\left(\frac{d}{dx} u(x)\right) u(x)\,x^2\,dx$$

We intend to apply the integration by parts for the second term of the above result

```
> e31:=int(4*u(x)^2*x,x = -1 .. 1);
```

$$e31 := \int_{-1}^{1} 4\,u(x)^2\,x\,dx$$

```
> e32:=int(4*diff(u(x),x)*u(x)*x^2,x = -1 .. 1);
```

$$e32 := \int_{-1}^{1} 4\left(\frac{d}{dx} u(x)\right) u(x)\,x^2\,dx$$

```
> e32:=intparts(e32,x^2);
```

$$e32 := 2\,u(1)^2 - 2\,u(-1)^2 - \int_{-1}^{1} 4\,u(x)^2\,x\,dx$$

Notice that in the expression of e32 the integral apears in black, which means that is seen as a inert integral (it is used **Int**). In order to reduce this integral in the expression of e3 we have to use the value of e32. We can do that using **value** command

```
> e3:=e31+value(e32);
```

$$e3 := 2\,u(1)^2 - 2\,u(-1)^2$$

```
>
> delta2:=e3;
```

$$\delta 2 := 2\,u(1)^2 - 2\,u(-1)^2$$

Because $u(1) = u(-1)$ $= 0$ then $\delta 2 = 0$